
1. Overview

This document lists a TTCN change needed to correct issues in the ‘IMS_36523_IWD_12wk48’ test suite which is included in ATS ‘iwd-EUTRA-B2012-03_D12wk48’.

3. Corrections

Change 1 – Correction to function ‘f_IMS_ReceiveRequest()’
	Function name
	f_IMS_ReceiveRequest ()

	Reason for change
	The value of 30 seconds in the IMS Receive Request procedure was too low to ensure that 
a) the user switches on the UE and triggers IMS registrations via MMI, 
b) the UE boots, performs cell selection and finally attaches to the cell.



	Summary of change
	Timer value increased to 180 sec. in function ‘f_IMS_ReceiveRequest()’.

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before change

	...

function f_IMS_ReceiveRequest(template IMS_DATA_REQ p_IMS_DATA_REQ,

                                float p_TimerVal := 30.0) runs on IMS_PTC return IMS_DATA_REQ // @olaf timout increasedd

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    timer t_Watchdog := p_TimerVal;

    t_Watchdog.start;

    IMS_Server.receive(p_IMS_DATA_REQ) -> value v_IMS_DATA_REQ;      // receive request

    t_Watchdog.stop;

    return v_IMS_DATA_REQ;

  }

...


After change

	...

  group SendReceiveFunctions {

  function f_IMS_ReceiveRequest(template IMS_DATA_REQ p_IMS_DATA_REQ,

                                float p_TimerVal := 180.0) runs on IMS_PTC return IMS_DATA_REQ // @olaf timout increasedd

  {

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    timer t_Watchdog := p_TimerVal;

    t_Watchdog.start;

    IMS_Server.receive(p_IMS_DATA_REQ) -> value v_IMS_DATA_REQ;      // receive request

    t_Watchdog.stop;

    return v_IMS_DATA_REQ;

  }

...
	…


Change 2 – Correction to function ‘f_MTC_IMS_MainLoop()’
	Function name
	f_MTC_IMS_MainLoop ()

	Reason for change
	The logic for the determination of done condition in the main while loop was not correct.


	Summary of change
	Logic corrected, see below.

	TTCN module
	IMS_34229\MTC_IMS_36523.ttcn

	MCC160 Comment
	


Before change

	...

  function f_MTC_IMS_MainLoop(timer p_GuardTimer) runs on MTC_IMS

  { /* MTC main loop; called from the testcases

     * NOTE: it is assumed that the IMS_PTC is started just before calling this loop */

    var boolean v_IMS1_Done := (ptc_IMS1 != null);

    var boolean v_IMS2_Done := (ptc_IMS2 != null);
    var template (omit) IMS_UT_CommandSource_Type v_CommandSource := omit;

    var boolean v_WaitingForCnf;

    var boolean v_CnfRequiredbyRef;

    while (not (v_IMS1_Done or v_IMS2_Done)) {
      v_WaitingForCnf := isvalue(v_CommandSource);

      alt {

        [not v_IMS1_Done] ptc_IMS1.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS1_Done := true;

        }

        [not v_IMS2_Done] ptc_IMS2.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS2_Done := true;

        }

        [not v_WaitingForCnf] a_MTC_ReceiveFromIMS_SendToUT(v_CnfRequiredbyRef, IMS_MMI[tsc_Index_PDN1], ims1) {

          if (v_CnfRequiredbyRef) {

            v_CommandSource := ims1;

          }

        }

        [not v_WaitingForCnf] a_MTC_ReceiveFromIMS_SendToUT(v_CnfRequiredbyRef, IMS_MMI[tsc_Index_PDN2], ims2) {

          if (v_CnfRequiredbyRef) {

            v_CommandSource := ims2;

          }

        }

        [not v_WaitingForCnf] a_MTC_ReceiveFromIPCAN_SendToUT(v_CnfRequiredbyRef) {

          if (v_CnfRequiredbyRef) {

            v_CommandSource := ipcan;

          }

        }

        [] a_MTC_ReceiveFromUT(v_CommandSource) {

          v_CommandSource := omit;

        }

        [] p_GuardTimer.timeout {

          f_MTC_IMS_Stop(fail);

        }

        [] any component.killed {

          f_MTC_IMS_Stop(inconc);

        }

      }

    }

    f_MTC_IMS_Stop(pass);   /* Note: "pass" overrules "none" only */

  }

}
...


After change

	...

  function f_MTC_IMS_MainLoop(timer p_GuardTimer) runs on MTC_IMS

  { /* MTC main loop; called from the testcases

     * NOTE: it is assumed that the IMS_PTC is started just before calling this loop */

    var boolean v_IMS1_Done := (ptc_IMS1 == null);

    var boolean v_IMS2_Done := (ptc_IMS2 == null);
    var template (omit) IMS_UT_CommandSource_Type v_CommandSource := omit;

    var boolean v_WaitingForCnf;

    var boolean v_CnfRequiredbyRef;

    while (not (v_IMS1_Done and v_IMS2_Done)) {
      v_WaitingForCnf := isvalue(v_CommandSource);

      alt {

        [not v_IMS1_Done] ptc_IMS1.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS1_Done := true;

        }

        [not v_IMS2_Done] ptc_IMS2.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS2_Done := true;

        }

        [not v_WaitingForCnf] a_MTC_ReceiveFromIMS_SendToUT(v_CnfRequiredbyRef, IMS_MMI[tsc_Index_PDN1], ims1) {

          if (v_CnfRequiredbyRef) {

            v_CommandSource := ims1;

          }

        }

        [not v_WaitingForCnf] a_MTC_ReceiveFromIMS_SendToUT(v_CnfRequiredbyRef, IMS_MMI[tsc_Index_PDN2], ims2) {

          if (v_CnfRequiredbyRef) {

            v_CommandSource := ims2;

          }

        }

        [not v_WaitingForCnf] a_MTC_ReceiveFromIPCAN_SendToUT(v_CnfRequiredbyRef) {

          if (v_CnfRequiredbyRef) {

            v_CommandSource := ipcan;

          }

        }

        [] a_MTC_ReceiveFromUT(v_CommandSource) {

          v_CommandSource := omit;

        }

        [] p_GuardTimer.timeout {

          f_MTC_IMS_Stop(fail);

        }

        [] any component.killed {

          f_MTC_IMS_Stop(inconc);

        }

      }

    }

    f_MTC_IMS_Stop(pass);   /* Note: "pass" overrules "none" only */

  }

}
...
	…


Change 3 – Correction to altstep ‘a_IP_IMS_UL_ReceiveDataUDP()’
	Altstep name
	a_IP_IMS_UL_ReceiveDataUDP()

	Reason for change
	The variables v_NW_Address and v_UE_Address were undefined in case the UE was not registered.


	Summary of change
	Variables v_NW_Address and v_UE_Address are set according to the header of the incoming IP_SOCKET_IND.

	TTCN module
	IP_PTC\IP_PTC_IMS_Handler.ttcn

	MCC160 Comment
	


Before change

	...

  altstep a_IP_IMS_UL_ReceiveDataUDP(IMS_Server_Type p_ImsServer,

                                     integer p_PdnIndex,

                                     template (present) IP_AddrInfo_Type p_IpAddr,

                                     template (present) PortNumber_Type  p_PortNumber,

                                     IMS_ProtectedUnprotected_Type p_ProtectedUnprotected) runs on IP_PTC

  { /* receive data from the UE */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var Datagram_Content_Type v_Data;

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var IMS_ClientServer_Type v_ClientServer;

    // receive UDP data

    [] IP_SOCK.receive(cr_UDP_DATA_IND(cr_IP_Socket(p_IpAddr, p_PortNumber)))

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom.Buffer;

        if (not fl_ImsServer_UeIsRegistered(p_ImsServer)) {

          /* initial access:  => IMS PTC gets handed over the UE's and the NW's IP address (=> IPv4 or IPv6) */

          v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;    // may be IPv4 or IPv6

          v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr;   // may be IPv4 or IPv6
          v_ClientServer := server;   // intial access is only applicable for requests from the UE

        }

        else {

          v_ClientServer := fl_IP_IMS_UL_ResponseOrRequest(v_Data);   /* for UDP Requests and Responses are sent by the UE from Port_uc to Port_ps,

                                                                       * i.e. there is not data for Port_us to Port_pc and therefore Requests and Responses

                                                                       * cannot be distinguished by the port on which they have been received */

        }

        v_RoutingInfo := cs_IMS_RoutingInfo(udp, p_ProtectedUnprotected, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, v_ClientServer, v_RoutingInfo, v_Data);

      }

  }

...


After change

	...

  altstep a_IP_IMS_UL_ReceiveDataUDP(IMS_Server_Type p_ImsServer,

                                     integer p_PdnIndex,

                                     template (present) IP_AddrInfo_Type p_IpAddr,

                                     template (present) PortNumber_Type  p_PortNumber,

                                     IMS_ProtectedUnprotected_Type p_ProtectedUnprotected) runs on IP_PTC

  { /* receive data from the UE */

    var IP_SOCKET_IND v_IP_SOCKET_IND;

    var Datagram_Content_Type v_Data;

    var template (omit) IP_AddrInfo_Type v_NW_Address := omit;

    var template (omit) IP_AddrInfo_Type v_UE_Address := omit;

    var template (value) IMS_RoutingInfo_Type v_RoutingInfo;

    var IMS_ClientServer_Type v_ClientServer;

    // receive UDP data

    [] IP_SOCK.receive(cr_UDP_DATA_IND(cr_IP_Socket(p_IpAddr, p_PortNumber)))

      -> value v_IP_SOCKET_IND

      {

        v_Data := v_IP_SOCKET_IND.DATA.Ind.UDP.RecvFrom.Buffer;


   v_NW_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Local.IpAddr;    

        v_UE_Address := v_IP_SOCKET_IND.DATA.ConnectionId.Remote.IpAddr;   
        if (not fl_ImsServer_UeIsRegistered(p_ImsServer)) {

          /* initial access:  => IMS PTC gets handed over the UE's and the NW's IP address (=> IPv4 or IPv6) */

          v_ClientServer := server;   // intial access is only applicable for requests from the UE

        }

        else {

          v_ClientServer := fl_IP_IMS_UL_ResponseOrRequest(v_Data);   /* for UDP Requests and Responses are sent by the UE from Port_uc to Port_ps,

                                                                       * i.e. there is not data for Port_us to Port_pc and therefore Requests and 
Responses

                                                                       * cannot be distinguished by the port on which they have been received */

        }

        v_RoutingInfo := cs_IMS_RoutingInfo(udp, p_ProtectedUnprotected, v_UE_Address, v_NW_Address);

        fl_IP_IMS_UL_DecodeAndSend(p_PdnIndex, v_ClientServer, v_RoutingInfo, v_Data);

      }

  }

...
	…

